# $Id$
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=cln
pkgver=1.3.4
pkgrel=3
pkgdesc="Class library for numbers"
arch=('i686' 'x86_64')
url="http://www.ginac.de/CLN/"
license=('GPL')
depends=('gmp')
makedepends=('texlive-core')
source=(http://www.ginac.de/CLN/${pkgname}-${pkgver}.tar.bz2)
sha1sums=('76f73071236ead72ba5c9ee892f29ca24e557b8c')

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr
  make all html pdf
}

check() {
  cd ${pkgname}-${pkgver}
  make check
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install install-html install-pdf
}
